

# 進階函數與資料處理

## 課程目標

- 掌握資料比對（搜尋/參照）、陣列運算與動態範圍設定技巧
- 熟悉進階文字處理與清理方法，能處理雜訊型資料

## 建立新工作表

1. 複製農產品交易紀錄表，**ctrl** 加拖曳農產品交易紀錄表。
2. 選擇一個市場後，將其它市場資料都刪除。
3. 選取任一儲存格後選擇**常用 > 格式化為表格**。
4. 點選**表格設計**功能，選擇移除重複項

## 文字處理函數

### 文字合併函數

#### CONCAT

- 定義：合併多個範圍或字串中的文字。此函式為 CONCATENATE 的現代取代者，主要優點是支援「範圍」選取。
- 使用語法：CONCAT(文字1, [文字2], ...)
- 實際範例：**=CONCAT(A2:C2)**。
  - 說明：假設 A1 是「台北市」、B1 是「信義區」、C1 是「市府路」，此公式會將 A1 到 C1 範圍內所有儲存格的內容合併，傳回 "台北市信義區市府路"。注意：CONCAT 不會在項目之間自動加入分隔符號。

#### TEXTJOIN

- 定義：合併多個範圍或字串中的文字，並且可以在每個項目之間插入指定的「分隔符號」，還能選擇是否「忽略空格」。是目前功能最強大、最有彈性的文字合併函式。
- 使用語法：TEXTJOIN(分隔符號, 是否忽略空格, 文字1, [文字2], ...)
- 實際範例：**=TEXTJOIN(", ", TRUE, A2:C2)**。
  - 說明：此公式會將 A1:A5 範圍內的所有文字合併。每個項目之間會用「, 」(逗號和空格) 作為分隔符號。TRUE：代表如果範圍中有空白儲存格，將會被忽略。例如，若 A1:A3 為 "蘋果", "橘子", "香蕉"，結果會是 "蘋果, 橘子, 香蕉"。

實作練習：在員工資料表新增一個欄位，將姓名、部門及月薪合併成一筆資料，並且以", "隔開。

```
=TEXTJOIN(", ", TRUE, A2:C2, F2)
```

## 文字分析函數

### LEN

- 定義：傳回文字字串的「字元」個數。
- 使用語法：LEN(文字字串)
- 實際範例：`=LEN("Excel 學習")`。
  - 說明：計算 "Excel 學習" 的字元總數。E,x,c,e,l, ,學,習 共 8 個字元，因此公式會傳回 8。
  - 相關函式：LENB 則是計算「位元組(Byte)」數，一個中文字會算成 2，所以 `=LENB("Excel 學習")` 會傳回 10 (6個英數字元 + 2\*2個中文字元)。

---

### LEFT

- 定義：從一個文字字串的開頭 (最左邊) 起，傳回指定數量的「字元」。
- 使用語法：LEFT(文字字串, [取幾個字])
- 實際範例：`=LEFT("台北市信義區", 3)`。
  - 說明：傳回「台北市信義區」這個字串中最左邊的 3 個字元，結果為 "台北市"。

### RIGHT

- 定義：從一個文字字串的結尾 (最右邊) 起，傳回指定數量的「字元」。
- 使用語法：RIGHT(文字字串, [取幾個字])
- 實際範例：`=RIGHT("SKU-2025-XYZ", 3)`。
  - 說明：傳回「SKU-2025-XYZ」這個字串中最右邊的 3 個字元，結果為 "XYZ"。

### MID

- 定義：從一個文字字串的中間，於指定起始位置起，傳回指定數量的「字元」。
- 使用語法：MID(文字字串, 開始, 取幾個字)
- 實際範例：`=MID("SKU-2025-XYZ", 5, 4)`。
  - 說明：從「SKU-2025-XYZ」這個字串的第 5 個字元 (2) 開始，擷取 4 個字元，結果為 "2025"。

實作練習：開啟 job104.xlsx 在欄位地區後面新增兩欄分別為市跟區，將地去分割後分別填入。

```
=MID(C2, 1, 3)
=IF(MID(C2, 4, 3)="", "無", MID(C2, 4, 3))
```

### FIND

- 定義：在一個文字字串中尋找另一個文字字串，並傳回其起始位置的編號。此函式會區分英文大小寫，且不支援萬用字元。
- 使用語法：FIND(查詢字串, within\_text, [start\_num])
- 實際範例：`=FIND("-", "SKU-2025-XYZ")`。
  - 說明：在「SKU-2025-XYZ」字串中尋找第一個「-」出現的位置，結果為 4。若使用 `=FIND("xyz",`

"SKU-2025-XYZ")

則會因大小寫不符而傳回 #VALUE! 錯誤。

## SEARCH

- 定義：在一個文字字串中尋找另一個文字字串，並傳回其起始位置的編號。此函式**不區分英文大小寫**，且支援萬用字元 ( ? 和 \* )。
- 使用語法：SEARCH(find\_text, within\_text, [start\_num])
- 實際範例：=SEARCH("xyz", "SKU-2025-XYZ")。
  - 說明：在「SKU-2025-XYZ」字串中尋找 "xyz"。因為 SEARCH 不區分大小寫，所以它會成功找到大寫的 "XYZ"，並傳回其起始位置 10。

## 文字轉換函數

### UPPER

- 定義：將文字字串中的所有小寫英文字母轉換成「大寫」。
- 使用語法：UPPER(文字字串)
- 實際範例：=UPPER("Hello world")。
  - 說明：將「Hello World」這個字串轉換成全大寫，結果為 "HELLO WORLD"。

### LOWER

- 定義：將文字字串中的所有大寫英文字母轉換成「小寫」。
- 使用語法：LOWER(文字字串)
- 實際範例：=LOWER("Hello world")。
  - 說明：將「Hello World」這個字串轉換成全小寫，結果為 "hello world"。

### PROPER

- 定義：將文字字串中每個英文單字的「字首」轉換為大寫，其餘字母則轉換為小寫。
- 使用語法：PROPER(文字字串)
- 實際範例：=PROPER("welcome to taipei")。
  - 說明：將「welcome to taipei」這個字串轉換成字首大寫格式，結果為 "Welcome To Taipei"。

### TRIM

- 定義：移除文字字串中多餘的「空格」，只保留單字之間的一個空格。常用於清理從外部匯入、帶有不規則空格的資料。
  - 使用語法：TRIM(文字字串)
  - 實際範例：=TRIM(" Hello world ")。
    - 說明：移除字串前後所有空格，並將字串中間的多個空格壓縮成一個，結果為 "Hello World"。
-

## REPLACE

- 定義：根據指定的「起始位置」與「長度」，將文字字串的一部分替換成另一個不同的文字字串。
- 使用語法：REPLACE(文字字串, 從第幾個字, 修改字數, 替換字串)
- 實際範例：=REPLACE("ABC-123-XYZ", 5, 3, "456")。
  - 說明：在「ABC-123-XYZ」字串中，從第 5 個字元 (1) 開始，將 3 個字元 (123) 替換成 "456"，結果為 "ABC-456-XYZ"。

實作練習：將農產品交易行情的種類代碼 N05 改為 NO-05

```
=REPLACE(B2, 1, 1, "NO-")
```

## SUBSTITUTE

- 定義：在文字字串中，將指定的「舊文字」替換成「新文字」。您可以選擇要替換第幾個出現的舊文字。
- 使用語法：SUBSTITUTE(文字字串, 替換字串, 新字串, [第幾個])
- 實際範例：=SUBSTITUTE("2025/10/15", "/", "-")。
  - 說明：在「2025/10/15」字串中，將所有出現的 / 都替換成 -，結果為 "2025-10-15"。
- 進階範例：=SUBSTITUTE("A-B-C-D", "-", "\*", 2)。
  - 說明：只將第 2 個出現的 - 替換成 \*，結果為 "A-BC-D"。

## TEXTSPLIT

- 定義：使用指定的分隔符號，將文字字串「分割」成多個部分，並將結果動態展開到多個儲存格中 (可以是橫向一列或縱向一欄)。這是取代傳統「資料剖析」功能的現代函式。
- 使用語法：TEXTSPLIT(文字, 欄分隔符號, [列分隔符號], [是否忽略空格], ...)
- 實際範例 1 (分割至不同欄)：=TEXTSPLIT("蘋果,橘子,香蕉", ",", "")
  - 說明：此公式會使用「,」(逗號) 作為欄分隔符號，將文字 "蘋果,橘子,香蕉" 分割。結果會自動向右「溢出」到三個相鄰的儲存格中，分別顯示「蘋果」、「橘子」、「香蕉」。
- 實際範例 2 (分割至不同列)：=TEXTSPLIT("蘋果;橘子;香蕉", ";", "")
  - 說明：此公式省略了「欄分隔符號」(第二個參數留空)，並使用「;」(分號) 作為列分隔符號。結果會自動向下展開到三個儲存格中。

如何使用函式處理薪資範圍後分別放入欄位B及欄位C?

使用 TEXTSPLIT 函數，非常簡潔高效。您只需要在 B1 儲存格輸入一個公式，它會自動將兩個結果「溢出 (Spill)」到 B1 和 C1。

步驟：

1. 先用 TEXTAFTER 和 TEXTBEFORE 去掉頭尾的文字。
2. 再用 TEXTSPLIT 以 ~ 符號為分隔符，將文字拆開。
3. 最後用 SUBSTITUTE 去掉千分位逗號，並用兩個負號 -- 將文字轉換為數字。

在 B1 儲存格輸入以下公式：

```
1 =--SUBSTITUTE(TEXTSPLIT(TEXTBEFORE(TEXTAFTER(A1,"月薪"),"元"),"~"),",","")
```

公式解析：

- TEXTAFTER(A1,"月薪"): 取得 "月薪" 後面的文字，結果為 "32,000~39,000元"。
- TEXTBEFORE(...): 從上一步的結果中，取得 "元" 前面的文字，結果為 "32,000~39,000"。
- TEXTSPLIT(...,"~"): 將上一步的結果用 "~" 拆開，得到 {"32,000", "39,000"} 這樣一個陣列。Excel 會自動將這兩個值分別放入 B1 和 C1。
- SUBSTITUTE(...,"",""): 去掉陣列中每個值的逗號，得到 {"32000", "39000"}。
- --: 這是將文字轉換為數字的速記法，等同於 VALUE() 函數。最終得到 {32000, 39000}。

## 動態陣列函數

### UNIQUE

- 定義：從一個指定的範圍或陣列中，傳回一組不重複的「唯一值」清單。這是一個「動態陣列 (Dynamic Array)」函式，其結果會自動「溢出 (Spill)」到下方的儲存格中，無需手動向下拖曳公式。
- 使用語法：UNIQUE(範圍, [依欄位比較], [只出現一次])
  - 範圍 (array): 您要從中擷取唯一值的來源範圍。
  - [依欄位比較] (by\_col): (選填) 預設為 FALSE (逐列比較)。若設為 TRUE 則會變成逐欄比較。
  - [只出現一次] (exactly\_once): (選填)
    - FALSE (預設): 傳回所有不重複的項目 (例如 A,B,A,C 會傳回 A,B,C)。
    - TRUE: 只傳回在來源中僅出現過一次的項目 (例如 A,B,A,C 只會傳回 B,C)。
- 實際範例:

範例 1：取得不重複的清單 (預設用法)

情境：從 A 欄的產品清單中，建立一份不重複的產品列表。

公式：=UNIQUE(A2:A9)

- 說明：這是最常見的用法。函式會掃描 A2:A9，找出所有不重複的產品名稱，並在您輸入公式的儲存格及其下方自動建立一個新的清單。
- 結果 (會自動溢出到四個儲存格)：

範例 2：只傳回「出現過一次」的項目

情境：從 A 欄的產品清單中，找出哪些產品是只被記錄過一次的。

公式：=UNIQUE(A2:A9, , TRUE)

- 說明：注意我們省略了第二個引數，並將第三個引數設為 TRUE。函式會找出在來源清單中「從頭到尾只出現過一次」的項目。「蘋果」和「橘子」都出現了兩次，所以被排除了。

練習: 在員工資料表列出公司所有部門及每個部門有幾名員工

```
=UNIQUE(B2:B191)
=COUNTIF(B:B,J2)
```

## SEQUENCE

- 定義：產生一組連續的「數列」，例如 1, 2, 3...。可以自由指定要產生的「列數」、「欄數」、起始值和間距。
- 使用語法：SEQUENCE(列數, [欄數], [起始值], [間距])
- 實際範例 1：=SEQUENCE(10)
  - 說明：產生一個 10 列 1 欄的數列，從 1 到 10。
- 實際範例 2：=SEQUENCE(3, 4, 10, 5)
  - 說明：產生一個 3 列 4 欄的矩陣。數列從 10 開始，每個數字之間的時間距為 5 (即 10, 15, 20, ...)。

## FILTER

- 定義：根據指定的「條件」，從一個資料範圍中「篩選」出所有符合條件的紀錄。這是一個動態陣列函式，結果會自動溢出。
- 使用語法：FILTER(要篩選的範圍, 條件, [如果沒有結果時])
- 實際範例：=FILTER(A2:C10, A2:A10="業務部", "查無資料")
  - 說明：從 A2:C10 範圍中，篩選出所有 A 欄等於「業務部」的紀錄。如果找不到任何紀錄，則顯示「查無資料」。

練習: 在農產品交易行情取出所有極柑交易資料

## HSTACK

- 定義：將多個範圍或陣列「水平地 (Horizontally)」堆疊在一起，組合成一個更寬的新陣列。
- 使用語法：HSTACK(範圍1, [範圍2], ...)
- 實際範例：=HSTACK(A1:A5, D1:D5)
  - 說明：將 A1:A5 範圍和 D1:D5 範圍的資料，左右並排成一個新的表格。結果會是一個 5 列 2 欄的陣列。
  - 相關函式：其對應的函式是 VSTACK，用於「垂直地」堆疊資料。

練習: 在農產品交易行情產生一個新表格，欄位為作物名稱，平均價，交易量及總成交值

## SORT

- 定義：對一個範圍或陣列的內容進行「排序」。它會傳回一個已排序的新陣列，而不會改動到您的原始資料。
- 使用語法：  
SORT(範圍, [排序索引], [排序順序], [按欄排序])
- 實際範例 1：=SORT(A2:A10)
  - 說明：對 A2:A10 範圍的內容進行預設的「遞增」排序 (A-Z 或小到大)。

- 實際範例 2：=SORT(A2:C10, 3, -1)

- 說明：對 A2:C10 整個範圍進行排序。排序的依據是第 3 欄 (C 欄)。排序的順序為 -1，代表「遞減」排序 (大到小)。

## 搜尋與參照函數

課程內容：

- LOOKUP 函數家族介紹與比較

- VLOOKUP：垂直搜尋 (左到右)，搜尋欄位必須在範圍最左側
- HLOOKUP：水平搜尋 (上到下)，用法與 VLOOKUP
- XLOOKUP (Excel 365 可選)：功能完整，支援反向搜尋、未找到回傳值、精確/近似參照

### VLOOKUP (垂直查詢)

- 定義：在表格的「最左欄」中進行「垂直查詢」，並傳回同一列中指定欄位的數值。
- 使用語法：VLOOKUP(查詢值, 表格範圍, 欄位索引, [是否完全符合])
  - 查詢值：要比對的關鍵欄位，如 ID 或編號
  - 表格範圍：包含要查詢的值及要擷取資料的完整表格範圍
  - 欄位索引：從左數過來第幾欄資料要被擷取 (數字 1,2,3...)
  - 是否完全符合：通常使用 FALSE(精確比對)以避免錯誤。
- 實際範例：=VLOOKUP(C5,C2:F19,4,FALSE)。
  - 說明：在員工資料表 C2:F19 的範圍中，尋找最左欄 (C欄) 中值為 C5 的儲存格，找到後，傳回該列的第 4 欄 (F欄) 的值。FALSE 代表要求「完全符合」。

檔案

常用

插入

共用

版面配置

公式

資料

檢閱

檢視

說明

繪圖

↶

▼

📄

▼

🖌

▼

新細明體 (本文)

▼

12

▼

B

📏

▼

🖌

▼

🔴

▼

⋯

▼

📏

▼

🔗

▼

📏

▼

通用格式

▼

\$€

G5

▼

✕

✓

fx

▼

=VLOOKUP(C5,C2:F19,4,FALSE)

	A	B	C 1	D 2	E 3	F 4	G
1	姓名	部門	電話	到職日期	到職日期	月薪	
2	王冠偉	工程部	918204352	43809	2019/12/10	54190.00	
3	陳文俊	行政部	912820495	45627	2024/12/1	37713.00	
4	張明君	業務部	92868476	44655	2022/4/4	78336.00	
5	劉志育	財務部	933954062	43998	2020/6/8	33793.00	33793
6	林怡哲	客服部	976298045	44873	2022/11/8	36616.00	
7	黃冠智	工程部	979072364	44379	2021/7/2	66421.00	
8	鄭嘉美	研發部	986309822	43508	2019/2/12	73409.00	
9	蔡宏臻	客服部	919307823	45106	2023/6/29	61393.00	
10	李志興	人資部	971855047	45007	2023/3/22	56387.00	
11	吳芳哲	工程部	938324029	45342	2024/2/20	109926.00	
12	王怡美	研發部	916342155	43770	2019/11/1	103549.00	
13	陳明蓉	採購部	913959176	43646	2019/6/30	72347.00	


### HLOOKUP (水平查詢)

- 定義：在表格的「最頂列」中進行「水平查詢」，並傳回同一欄中指定列位的數值。
- 使用語法：HLOOKUP(查詢值, 表格範圍, 列位索引, [是否完全符合])
  - 查詢值：您要查詢的值 (例如 "三月")。
  - 表格範圍：包含資料的整個範圍 (例如 A1:D5)。

- 列位索引: 您想從中取值的列號。在表格範圍中，第一列是 1，第二列是 2，依此類推。
- [是否完全符合]:
  - FALSE 或 0：要求完全符合的查詢值。(最常用)
  - TRUE 或 1：允許近似符合的查詢值。使用此選項時，第一列的值必須是遞增排序。

● 實際範例：=HLOOKUP("平均價",D1:K11,4,FALSE)

- 說明：在 D1:K11 的範圍中，尋找最頂列 (第1列) 中值為 "平均價" 的儲存格，找到後，傳回該欄的第 4 列 (第4列) 的值。FALSE 代表要求「完全符合」。



	A	B	C	D	E	F	G	H	I	J	K	L
1	交易日期	種類代碼	作物代碼	作物名稱	市場代號	市場名稱	上價	中價	下價	平均價	交易量	
2	114.10.14	N05	11	椰子	104	台北二	29.3	22.5	16.1	22.6	1032	107.3
3	114.10.14	N05	129	椰子-剝殼進口	104	台北二	42.9	25.4	23.3	28.3	1550	
4	114.10.14	N05	31	釋迦	104	台北二	110	66.4	51.7	72.2	919	
5	114.10.14	N05	469	藍莓-進口	104	台北二	86.6	55.8	107.3	2320		
6	114.10.14	N05	50	百香果-其他	104	台北二	77	73.8	45.7	68.8	248	
7	114.10.14	N05	51	百香果-改良種	104	台北二	164.8	77.2	46.5	88.6	4805	
8	114.10.14	N05	70	小番茄-其他	104	台北二	300	298.3	280	295	24	

實作練習：新增員工資料表2，欄位為姓名、部門、電話及月薪，隨機複製五個姓名，利用 VLOOKUP 取得對應資料值

更簡潔的方式

解說：

1. {2, 3, 6} 我們用大括號 {} 將想回傳的多個欄位索引包起來，形成一個「水平陣列常數」。
2. 運作方式：當 VLOOKUP 找到 A2 對應的資料列後，它會看到您要求回傳第 2、第 3、和第 6 欄的資料。
3. 自動溢出 (Spill)：在動態陣列的支援下，Excel 會將這三個欄位的結果，自動「溢出」到您輸入公式的儲存格以及右側相鄰的兩個儲存格中。

## XLOOKUP

- 定義：新世代的萬用查詢函式，可以在一個範圍或陣列中尋找項目，並傳回對應的結果。它能取代 VLOOKUP 和 HLOOKUP，且功能更強大、使用更靈活。
- 使用語法：XLOOKUP (查詢值, 查詢範圍, 回傳範圍, [找不到時], [比對模式], [搜尋模式])
  - [找不到時] ([if\_not\_found])
    - 選用參數
    - 定義：如果 XLOOKUP 在 查詢範圍 中找不到 查詢值，它會回傳這個參數所指定的值。
    - 範例："查無此人" 或 0。
    - 重點提示：
      - 如果省略此參數，當找不到時，XLOOKUP 會回傳 #N/A 錯誤。
      - 這個參數取代了過去需要用 IFERROR 函數來包裝 VLOOKUP 的做法，讓公式更簡潔。



○ [比對模式] ([match\_mode])

- 選用參數
- 定義：指定 XLOOKUP 如何進行比對。
- 可選值：
  - 0 (預設值)：精確比對 (Exact match)。這是最常用的模式，要求 查詢值 必須與 查詢範圍 中的項目完全相同。
  - -1：精確比對或下一個較小項目 (Exact match or next smaller item)。如果找不到精確比對，則回傳下一個比 查詢值 小的項目。常用於查找級距（例如根據分數查找等級，或根據金額查找折扣）。
  - 1：精確比對或下一個較大項目 (Exact match or next larger item)。如果找不到精確比對，則回傳下一個比 查詢值 大的項目。
  - 2：萬用字元比對 (Wildcard match)。允許您在 查詢值 中使用 \* (代表任意多個字元) 和 ? (代表單一字元) 進行模糊比對。
- 重點提示：
  - 預設為 0 (精確比對)，這比 VLOOKUP 的預設值（近似比對）更安全，減少了錯誤發生的機率。
  - 使用 -1 或 1 時，查詢範圍 建議先進行排序，以確保結果的準確性。

○ [搜尋模式] ([search\_mode])

- 選用參數
- 定義：指定 XLOOKUP 在 查詢範圍 中搜尋的方向。
- 可選值：
  - 1 (預設值)：從第一個項目開始搜尋 (Search from first to last)。這是最常見的搜尋方向。
  - -1：從最後一個項目開始搜尋 (Search from last to first)。非常有用！例如，當您想找到某個員工的「最新」一筆紀錄時。
  - 2：二分搜尋 (遞增排序) (Binary search, ascending sort)。這是一種非常快速的搜尋演算法，但要求 查詢範圍 必須是遞增排序的。如果資料量非常大，可以顯著提升效能。
  - -2：二分搜尋 (遞減排序) (Binary search, descending sort)。同樣要求 查詢範圍 必須是遞減排序的。
- 重點提示：
  - 預設為 1。
  - 當您需要查找「最新」或「最後」一筆符合條件的紀錄時，將此參數設為 -1 會非常方便。
  - 使用 2 或 -2 時，請務必確認 查詢範圍 已經排序，否則會回傳錯誤的結果。

- 實際範例：=XLOOKUP(A2,員工資料表!A2:A40,員工資料表!F2:F40)。

- 說明：在 員工資料表!A2:A40 (查詢範圍) 中尋找 "A2"，並從 員工資料表!F2:F40 (回傳範圍) 中傳回同一個位置的值。XLOOKUP 預設為「完全符合」，且查詢與回傳範圍是分開的，使用上更安全直觀，也沒有 VLOOKUP 只能查最左欄的限制。

實作練習：在員工資料表2，增加一個欄位調薪，每個員工增加 4%

## MATCH (比對)

- 定義：在一個單行或單列的範圍中搜尋指定的項目，並傳回該項目在範圍中的「相對位置」(是第幾個)。
- 使用語法：MATCH(查詢值, 查詢範圍, [比對類型])
- 實際範例：`=MATCH("蔡宏蓉",A2:A12,0)`。
  - 說明：在 A1:A20 的範圍中，尋找 "張經理" 這個文字。如果 "張經理" 位於 A15 儲存格，此公式就會傳回 15。0 代表要求「完全符合」。

## INDEX (索引)

- 定義：傳回表格或範圍中，指定「列號」與「欄號」交叉位置的儲存格內容。它的作用就像是從地圖中根據座標找出對應的位置。
- 使用語法：INDEX(範圍, 列號, [欄號])
- 實際範例：`=INDEX(E2:E12, 8)`。
  - 說明：在 E1:E12 的範圍中，找出第 8 列，也就是 E8，並傳回其內容。

## INDEX + MATCH (組合應用)

這兩個函式組合起來，是 Excel 中取代 VLOOKUP 的經典、強大且靈活的查詢方法。

- 定義：MATCH 負責找到目標在哪一「列」，INDEX 則根據 MATCH 傳回的列號，從您指定的結果範圍中取出對應的值。
- 使用語法：`=INDEX(回傳值的範圍, MATCH(查詢值, 查詢值的範圍, 0))`
- 實際範例：`=INDEX(E2:E12, MATCH("蔡宏蓉",A2:A12,0))`。
  - 說明：
    1. MATCH("蔡宏蓉",A2:A12,0) 會先在 A 欄中找到 "蔡宏蓉" 的位置 (8)。
    2. 公式變成 =INDEX(E2:E12, 8)。
    3. INDEX 接著會去 E 欄中，找出第 8 個位置的值並傳回。
  - 這個組合沒有 VLOOKUP 只能從最左欄查詢的限制，更為靈活。

實作練習：使用 INDEX + MATCH 在農產品交易行情表取得香蕉的交易量

## 實作練習 1：用 VLOOKUP 與 INDEX+MATCH 對同一筆查詢做比較

目的：理解 VLOOKUP 受限於查詢欄位置，而 INDEX+MATCH 更靈活。

步驟：

1. 準備測試資料

員工編號	部門	總銷售
1001	行銷	5000
1002	財務	4000
1003	行銷	6000

## 2. 使用 VLOOKUP 查詢

- 公式：`=VLOOKUP(1002, A2:C4, 3, FALSE)`
- 說明：查找員工編號 1002，回傳第 3 欄（總銷售）
- 結果：4000

## 3. 使用 INDEX+MATCH 查詢

- 公式：`=INDEX(C2:C4, MATCH(1002, A2:A4, 0))`
- 說明：MATCH 找到 1002 在 A2:A4 的位置，再用 INDEX 回傳 C 欄對應值
- 結果：4000

## 4. 觀察欄位順序變動

- 將「總銷售」欄移到最左邊
- 再試一次 VLOOKUP：`=VLOOKUP(1002, A2:C4, 3, FALSE)`
  - 結果可能錯誤，因為 VLOOKUP 只能向右查詢
- INDEX+MATCH 仍正確：`=INDEX(C2:C4, MATCH(1002, B2:B4, 0))`

結論：VLOOKUP 受限於查詢欄在左側；INDEX+MATCH 跨欄穩健，適合欄位順序會變動的情況。

# 實作練習 2：使用 INDEX+MATCH 完成多條件查找

目的：實作「同時比對員工編號與部門」的查找，並用 IFERROR 避免 #N/A

步驟：

## 1. 準備測試資料

員工編號	部門	總銷售
1001	行銷	5000
1002	財務	4000
1003	行銷	6000
1002	行銷	3000

## 2. 建立多條件查找公式

- 公式（需 Ctrl+Shift+Enter）：

```
=IFERROR(  
    INDEX(C2:C5, MATCH(1, (A2:A5=1002)*(B2:B5="行銷"), 0)),  
    "無資料"  
)
```

○ 說明：

1. (A2:A5=1002) 會生成一個 TRUE/FALSE 陣列
2. (B2:B5="行銷") 也生成 TRUE/FALSE 陣列
3. 兩個陣列相乘 → 只有同時符合兩條件的位置為 1
4. MATCH 找到 1 的位置，INDEX 回傳對應總銷售
5. IFERROR 若找不到，回傳「無資料」

### 3. 結果驗證

- 查找員工編號 1002 且部門「行銷」→ 回傳 3000
- 查找員工編號 1001 且部門「財務」→ 回傳「無資料」

### 4. 可擴展操作

- 改查其他條件組合
- 將公式拖拉應用到其他查詢列